United States
Department of
Agriculture

Forest Service

Forest
Products
Laboratory

General
Technical
Report
FPL-GTR-63

Computer Programs for Adjusting the Mechanical Properties of 2-Inch Dimension Lumber for Changes in Moisture Content

James W . Evans
Jane K. Evans
David W. Green



Abstract

This paper presents computer programs for adjusting the mechanical properties of 2-in. dimension lumber for changes in moisture content. Mechanical properties adjusted are modulus of rupture, ultimate tensile stress parallel to the grain, ultimate compressive stress parallel to the gain, and flexural modulus of elasticity. The models are valid for moisture contents from 8 to. 23 percent (assumed green value). Although test results can be adjusted for moisture content, such adjustments decrease in accuracy with increasing change in moisture content. For this reason, the specimens should be conditioned as closely as possible to the target moisture content prior to test.

Keywords: Computer program, moisture content, mechanical properties, lumber

Language

ANSI Standard Fortran 77

February 1990

Evans, James W.; Evans, Jane K.; Green, David W. 1990. Computer programs for adjusting the mechanical properties of 2-inch dimension lumber for changes in moisture content. Gen. Tech. Rep, FPL-GTR-63. Madison, WI: U.S. Department of Agriculture, Forest Service; Forest Products Laboratory. 18 p.

A limited number of free copies of this publication are available to the public from the Forest Products Laboratory, One Gifford Pinchot Drive, Madison, WI 53705-2398. Laboratory publications are sent to more than 1,000 libraries in the United States and elsewhere.

The Forest Products Laboratory is maintained in cooperation with the University of Wisconsin.

Computer Programs for Adjusting the Mechanical Properties of 2-Inch Dimension Lumber for Changes in Moisture Content

James W. Evans, Supervisory Mathematical Statistician
Jane K. Evans, Computer Programmer Analyst
David W. Green, Research General Engineer

Description and Purpose

1. Background

Lumber strength properties vary with changes in moisture content. This is recognized and addressed in standards such as ASTM D 245-84 (ASTM 1987) through the use of factors that adjust lumber properties for changes in moisture content. Recent studies have established that the D 245 procedures for adjusting lumber properties for changes in moisture content are not necessarily accurate or conservative. Details of this research effort are discussed in several references (Aplin and others 1986; Green and others 1990; Jessome and Bellusillo 1985; McLain and others 1984) and summarized by Green and Evans (1989). The objective of this paper is to present a series of computer programs that are based on these recent findings and may be used to adjust lumber properties. Matched sets of Douglas-fir and Southern Pine lumber of different sizes and grades were tested at four moisture contents (10, 15, and 20 percent and green) in several studies. Changes in the properties of modulus of rupture (MOR), ultimate tensile strength (UTS), ultimate compressive stress (UCS), and flexural modulus of elasticity (MOE) were modeled as functions of change in moisture content (Green and others 1986, 1990, in preparation a,b).[1] The resulting models are incorporated in the computer programs listed in the Appendix and discussed in this paper. Although test results can be adjusted for moisture content, such adjustments decrease in accuracy with increasing change in moisture content. Specimens should therefore be conditioned as closely as possible to the target moisture content prior to test.

2. Description of the Algorithm

Two different types of algorithms that predict how lumber properties change as moisture content changes are incorporated in the programs-one algorithm for MOR, UTS, and UCS and another algorithm for MOE. The way each of these algorithms works is discussed in detail, beginning with the more complicated algorithm used for MOR, UTS, and UCS. Because the algorithm is basically the same for MOR, UTS, and UCS, we will use MOR to explain the algorithm.

a. Algorithm for Modulus of Rupture
Data resulting from the moisture content-strength

[1] Modeling of the UCS-moisture content relationships was conducted by Dr. J. D. Barrett, University of British Columbia, Vancouver, following procedures similar to those used in the publications cited.



Figure 1—Effect of moisture content on the modulus of rupture of 2 by 4 No. 2 Southern Pine. (Percentiles shown from bottom to top are 5, 10, 25, 50, 75, 90, and 95.)

studies were plotted for each size-grade cell of data to show the percentiles of the distributions of strength properties sampled at the four moisture content levels. Figure 1 is an example showing the 5, 10, 25, 50, 75, 90, and 95 percentile MOR values for each of four moisture contents for No. 2 grade Southern Pine 2 by 4's. Similar plots for other species, properties, and size-grade cells are presented in Green and Evans (1989).

Looking at specific percentiles across the four moisture content levels, we noted that the moisture content-strength relationship followed a quadratic-type curvature. For the lower percentiles, the curvature was usually concave downward, and for higher percentiles, concave upward. Looking at different size-grade cells, we noted that the curvature and direction of curvature were similar in the different cells for a given strength level. This apparent pattern led to the first assumption used to develop the model: for any of the contours, the strength S of the specimen on the contour can be represented by the quadratic equation

$$S = B_0 + B_1 M + B_{11} M^2 \qquad (1)$$

where M is moisture content and B_0, B_1, and B_{11} are coefficients of the quadratic equation. Using this assumption, the difference in strength S_1 at moisture content M_1 and strength S_2 at M_2 on the same contour would be

$$S_2 - S_1 = (B_0 + B_1 M_2 + B_{11} M_2^2) - (B_0 + B_1 M_1 + B_{11} M_1^2)$$

Rearranged, this becomes

$$S_2 = S_1 + B_1(M_2 - M_1) + B_{11}(M_2^2 - M_1^2) \qquad (2)$$

Equation (2) predicts how strength S_1 at any moisture content M_1 changes when going to a different moisture content M_2 on the contour. If the values of B_1 and B_{11} were known for the contour, the model would be complete. Because the values of B_1 and B_{11} are not known, they must be modeled. Graphing the values of B_1 and B_{11} against strength led to the second assumption of the model: B_1 and B_{11} can be expressed as cubic functions of strength S.

$$B_1 = C_0 + C_1 S + C_2 S^2 + C_3 S^3 \qquad (3)$$

$$B_{11} = D_0 + D_1 S + D_2 S^2 + D_3 S^3 \qquad (4)$$

Coefficients C_0, C_1, C_2, C_3, D_0, D_1, D_2, and D_3 were estimated from the data in individual studies.

Initial models of this type proved quite successful in modeling the effect of moisture changes. However, the irreversibility of the model added a complication. For example, suppose the strength S_2 at moisture content M_2 is to be predicted when starting from strength S_1 at moisture content M_1. The value S_1 would be used to estimate B_1 and B_{11} in Equations (3) and (4). These values of B_1 and B_{11} with the values for S_1, M_1, and M_2 would produce an estimate of S_2 when substituted in Equation (2). A problem occurs when going from S_2 at M_2 back to M_1. Because B_1 and B_{11} would now be estimated using S_2 instead of S_1, the values of B_1 and B_{11} would be slightly different. This produces a slightly different value for S_1 than the initial value. The inability of the model to go back and forth on the same contour led to a further modification to assure reversibility. The solution to the problem of making the model reversible is also the reason why a computer program is needed to apply the model.

The solution to reversibility amounted to picking a reference moisture content level and using only strengths at that moisture content to determine the values of B_1 and B_{11}. We decided to let the strength at a moisture content of 15 percent determine the contours. Any other choice for a reference moisture content would have produced different model coefficients, but essentially the same contours. To find strength S_2 at moisture content M_2 when starting from strength S_1 at moisture content M_1, some additional steps are required. The first step is to find which contour S_1 is on. Using a reference moisture content of 15 percent means finding the strength S_{15} at a moisture content of 15 percent that, when adjusted to M_1, would produce S_1. From Equation (2) this means finding S_{15} such that

$$S_1 = S_{15} + B_1(M_1 - 15) + B_{11}(M_1^2 - 15^2) \quad (5)$$

Because B_1 and B_{11} are now defined based on S_5, substituting Equations (3) and (4), with S replaced by S_{15}, into Equation (5) gives

$$S_1 = S_{15} + (C_0 + C_1 S_{15} + C_2 S_{15}^3 + C_3 S_{15}^3)(M_1 - 15)$$

$$+(D_0 + D_1 S_{15} + D_2 S_{15}^2 + D_3 S_{15}^3)(M_1^2 - 15^2) \quad (6)$$

All variables in Equation (6) are known except S_{15}. Thus the resulting cubic equation must be solved for S_{15}. Given S_{15}, values of Bland B_{11} can be calculated. Finally, Equation (2) can be used to go from S_1 at M_1 to the appropriate S_2 at M_2.

This model, generally referred to as the fixed quadratic surface model, has proved very useful in modeling the effect of moisture content on MOR, UTS, and UCS. The best estimates of C_0, C_1, C_2, C_3, D_0, D_1, D_2, and D_3 found from the data in individual studies of these properties are given as the eight elements of the vector B in the subroutines that actually adjust the MOR, UTS, and UCS values.

b. Algorithm for Modulus of Elasticity

The adjustment procedure for MOE is substantially simpler. To go from MOE denoted E_1 at moisture content M_1 to MOE denoted E_2 at moisture content M_2, the ratio E_2/E_1 is assumed to be the same for all MOE values at M_1 and M_2, that is,

$$E_2/E_1 = \text{Constant percentage} \quad (7)$$

The MOE data in the studies indicated that this constant percentage was adequately modeled by

$$\text{Constant percentage} = (A + BM_2)/(A + BM_1) \quad (8)$$

Combining Equations (7) and (8) gives the basic form for the constant percentage model for MOE,

$$E_2 = E_1(A + BM_2)/(A + BM_1) \quad (9)$$

From the experimental data, $A = 1.8566$ and $B = -0.023722$ were determined to be the best values.

Program Description

1. General Structure

The programs were designed to run interactively. However, creating a batch program that could adjust a large number of specimens from one moisture content to another would require minimal work. Because the structure of the MOR, UTS, and UCS programs are virtually identical, we need discuss only the MOR and MOE program. Inserting subroutine TCALC or subroutine STCALC for subroutine RCALC and changing MOR to the appropriate abbreviation in all output would produce a program for UTS or UCS with all the correct parameter values and limits on the model (see Appendix for program listings). The program starts by asking the user for the following information:

1. Choice of adjusting MOR, MOE, or both
2. Appropriate combination of MOR and MOE (Note that the program requests the exact information it needs based on the selection of which property or properties to adjust. Free format input for all responses simplifies usage of the programs.)
3. If MOR is chosen, whether normalization is required and, if so, the normalizing constant to use (see discussion on normalization)
4. Current moisture content of the specimen to be adjusted
5. Moisture content to which the properties should be adjusted

Given this information, the program calls SUBROUTINE MCADJ, which calculates the adjusted property values and writes them in a table of results.

SUBROUTINE MCADJ is where the program could be modified to run in batch processing mode. A sequence of calls to this subroutine with appropriate information for each specimen would enable the program to adjust several observations in one pass. SUBROUTINE MCADJ is used to appropriately scale the properties for the parameter values used and to set up normalization calls to further routines if desired. SUBROUTINE MCADJ calls two subroutines, ECALC and RCALC.

SUBROUTINE ECALC adjusts the MOE value at one moisture content to the MOE at a different moisture content using the constant percentage model. This subroutine is self-contained.

SUBROUTINE RCALC adjusts MOR values at one moisture content to MOR values at a different moisture content using the fixed quadratic surface model. RCALC calls the subroutine CUBIC to find the roots of a cubic equation as needed in Equation (6).

2. Individual Subroutines

SUBROUTINE MCADJ(RMOR,EMOE,AMC,TMC, NORMAL,SPMEAN,RADJ,EADJ,RNADJ)

Formal Parameters-The formal parameters for MCADJ are shown in Table 1.

Auxiliary Algorithm-None

SUBROUTINE ECALC(EMOE,AMC,TMC,EADJ)

Formal Parameters-The formal parameters for ECALC are the same as for MCADJ (Table 1).

Auxiliary Algorithm-None

SUBROUTINE RCALC(RMOR,AMC,TMC,RADJ)

Formal Parameters-The formal parameters for RCALC are the same as for MCADJ (Table 1).

Auxiliary Algorithm-None

SUBROUTINE CUBIC(C,R15,RMOR)

Formal Parameters-The formal parameters for CUBIC are shown in Table 2.

3. Restrictions and Error Indication

The program has no known limitations that would cause an abnormal termination. It checks for errors, such as negative values in submission of values of MOR, MOE, and the normalizing constant, and writes out a warning message if one is found. Moisture contents are limited to between 23 percent (an assumed green moisture content) and 8 percent (an assumed lower bound of the model). Moisture contents outside this range are assumed to be at the closest boundary. A target moisture content to be adjusted to outside this range will result in a warning message and termination of the program. Finally, a request to adjust an abnormally small MOR can result in a warning to check the correctness of the values produced.

The MOR adjustment model is assumed reasonable within some built-in limits that are based on limits in the data. The largest MOR for which reasonable data were available was about 13.0×10^3 lb/in^2 at 15 percent moisture content. The program assumes that all contours with larger MOR values at 15 percent moisture content are parallel to the one passing through this point. Slightly different data limits are imposed on compression and tension. Likewise each property has a lower limit. The lowest contour that we have data to support will be gradually fanned down to a flat contour at zero strength. Because the last contour is essentially flat for MOR already, contours parallel to it provide essentially the same solution.

4. Normalization

For species with substantially different properties than those used to create the models, it may be advisable to “scale” property adjustments relative to those found in the Douglas-fir and Southern Pine moisture studies from which the models were created. With this scaling procedure, which is referred to as normalization, properties of weaker species are scaled up before entering the models. Similarly, adjusted properties are scaled down by the same factor used initially. Green and Evans (1989) discuss some possible scaling factors. The computer program uses a scaling procedure based on the mean strength of 2 by 4 Select Structural lumber at 15 percent moisture content. The procedure scales both the mean and spread of a new data set to match that found in the moisture studies creating the model. To use this type of normalization, a mean property at 15 percent moisture content for 2 by 4 Select

Structural lumber of the species being adjusted must be entered. This usually involves adjusting the needed size-grade cell of data to 15 percent unnormalized and using the mean of this adjusted data as the “normalizer” for all the data.

The programs are set up to allow normalization for bending. To normalize for tension and compression, a variable in the subroutine MCADJ must be changed. For MOR, XMEAN = 10.12045. For UTS, XMEAN = 7.45279. For UCS, XMEAN = 5.785. Additionally, in this subroutine the code under the corresponding comment cards (App., p. 12) must be changed for UTS as follows:

```
.
.
.
C
C           NORMALIZING THE MODELS
C
      ZMOR=RMOR*(XMEAN/SPMEAN)
C
C           NORMALIZING TO TARGET MOISTURE CONTENT
C
      CALL TCALC (ZMOR, AMC, TMC, ADJRN)
C
C           TAXING OUT THE NORMALIZATION FOR
C             TARGET MOISTURE CONTENT
C
      RNADJ=ADJRN*(SPMEAN/XMEAN)
      IF(RMOR .LE. 0.0) RNADJ=0.0
45    CONTINUE
.
.
.
```

For UCS, an identical change would be made, except the subroutine called would be STCALC.

Test Results

Tables 3 to 6 list test values for MOR, MOE, UTS, and UCS, respectively, that can be used to verify that the program is working.

Availability of Programs

Both source listings and compiled versions of the programs described in this publication are available on 5.25- or 3.5-in. diskettes for IBM and IBM-compatible microcomputers. Requests should be sent to

Engineering Properties of Wood
USDA Forest Service
Forest Products Laboratory
One Gifford Pinchot Drive
Madison, WI 53705-2398

along with a diskette of the size requested.

References

Aplin, E.N.; Green, D.W.; Evans, J.W.; Barrett, J.D. 1986. The influence of moisture content on the flexural properties of Douglas Fir dimension lumber. Res. Pap. FPL 475. Madison, WI: USDA Forest Serv., Forest Prod. Lab.

ASTM. 1987. Standard method for establishing structural grades and related allowable properties for visually graded lumber. ASTM D245-81. Philadelphia, PA: American Society for Testing and Materials.

Green, D.W.; Evans, J.W.. 1989. Moisture content-property relationships for dimension lumber: decisions for the future. In: In-Grade Testing of Structural Lumber; 1988 April 25-26; Madison, WI. Proceedings 47363. Madison, WI: Forest Products Research Society: 44-55.

Green, DW; Link, CL; DeBonis, AL; McLain, TE. 1986. Predicting the effect of moisture content on the flexural properties of Southern Pine dimension lumber. Wood Fiber Science. 18(1):134-156.

Green, D.W.; Evans, J.W.; Barrett, J.D.; Aplin, E.N. 1988. Predicting the effect of moisture content on the flexural properties of Douglas-fir dimension lumber. Wood Fiber Science. 20(1):107-131.

Green, D.W.; Pellerin, R.F.; Evans, J.W.; Kretschmann, D.E. 1990. Moisture content and the tensile strength of Douglas-fir dimension lumber. Res. Pap. FPL 497. Madison, WI: USDA Forest Serv., Forest Prod. Lab.

Green, D.W.; Evans, J.W.; Link, C.L (in preparation, a). Moisture content effects in lumber beams. For publication in Journal of Structural Engineering.

Green, D.W.; Evans, J.W.; Pellerin, R.F.(in preparation, b). Moisture content-tensile strength relationships for Douglas-fir dimension lumber. For publication in Wood Fiber Science.

Jessome, A.P.; Bellusillo, S.B. 1985. In-grade compression strength of Douglas-fir lumber. Final report of contract 03-50-10-006. Ottawa, ON, Canada: Forintek Canada Corp., Eastern Laboratory.

McLain, TE; DeBonis, AL; Green, DW; Wilson, F.J.; Link, C.L. 1984. The influence of moisture content on the flexural properties of Southern Pine dimension lumber. Res. Pap. FPL 447. Madison, WI: USDA Forest Serv., Forest Prod. Lab.

Table 1— Formal parameters for MCADJ, ECALC, and RCALC

Parameter	Type	Input or output	Content
RMOR	Real	Input	MOR of the specimen
EMOR	Real	Input	MOE of the specimen
AMC	Real	Input	Actual moisture content of the specimens
TMC	Real	Input	Target moisture content for the specimen
NORMAL	Integer	Input	1 if normalization is desired; 0 if normalization is not desired
SPMEAN	Real	Input	Normalization constant if normalization is desired
RADJ	Real	output	Adjusted unnormalized MOR value
EADJ	Real	output	Adjusted MOE value
RADJN	Real	output	Adjusted normalized MOR value

Table 2— Formal parameters for CUBIC

Parameter	Type	Input or output	Content
C	Real array (4 × 1)	Input	Coefficients of the cubic equation $C(4)R_{15}^3 + C(3)R_{15}^2 + C(2)R_{15} + C(1) = 0$
R15	Real	output	Solution of cubic equation and strength at 15 percent moisture content that solves Equation (6)
RMOR	Real	Input	Contains the MOR of the specimen

Table 3— MOR test values

Initial MOR ($\times 10^3$ lb/in^2)	Initial MC (percent)	Normalization		Target MC (percent)	Adjusted MOR ($\times 10^3$ lb/in^2)	Adjusted normalized MOR ($\times 10^3$ lb/in^2)
		Use	SPMEAN			
7.600	23	No		10	11.987	
8.300	30	Yes	9.000	12	12.519	12.513
12.000	6	Yes	11.000	27	(Ends with warning that 27 is above assumed green moisture content of 23 percent)	
12.000	6	Yes	11.000	23	7.342	7.464
12.000	23	No		6	(Ends with warning that 6 is below lower limit of 8, the assumed lower limit of the model)	
12.000	23	No		8	19.223	
1.000	19	No		8	1.026	

Table 4— MOE test values

Initial MOE ($\times 10^6$ lb/in^2)	Initial MC (percent)	Target MC[a] (percent)	Adjusted MOE ($\times 10^6$ lb/in^2)
1.500	23	15	1.717
3.000	12	19	2.683

[a]Lower limit, 8 percent; upper limit, 23 percent.

Table 5— UTS test values

Initial UTS (× 10^3 lb/in^2)	Initial MC (percent)	Normalization		Target MC[a] (percent)	Adjusted UTS (× 10^3 lb/in^2)	Adjusted normalized UTS (× 10^3 lb/in^2)
		Use	SPMEAN			
7.600	23	No		10	8.764	
10.000	23	No		8	11.044	
2.000	19	Yes	6.000	12	1.807	1.837
1.000	8	Yes	8.0	15	1.310	1.287
0.500	23	No		8	0.386	

[a] Lower limit, 8 percent; upper limit, 23 percent.

Table 6— UCS testvalues

Initial UCS (× 10^3 lb/in^2)	Initial MC (percent)	Normalization		Target MC[a] (percent)	Adjusted UCS (× 10^3 lb/in^2)	Adjusted normalized UCS (× 10^3 lb/in^2)
		Use	SPMEAN			
4.000	23	No		10	7.377	
7.000	23	No		8	11.730	
3.000	19	Yes	4.000	12	3.725	4.161
5.000	8	Yes	6.000	15	3.921	3.962
1.000	23	No		8	1.291	

“Lower limit, 8 percent; upper limit, 23 percent.

Appendix— Program Listings

1. MOR and MOE program

```
      WRITE(*,100)
100  FORMAT(1X, 'WELCOME TO THE SPECIES INDEPENDENT BENDING MOISTURE ADJ
    XUSTMENT PROGRAM')
      LOOP=0
 10  CONTINUE
      WRITE(*,105)
105  FORMAT(1X, 'DO YOU WANT TO ADJUST MOR, MOE, OR BOTH (TYPE 1=MOR, 2=
    XMOE, 0=BOTH)?')
      READ(*,*) IDRUN
 15  CONTINUE
      IF (IDRUN .NE. 0) GO TO 20
      WRITE(*,110)
110  FORMAT(1X, 'PLEASE ENTER MOE VALUE')
      READ(*,*) EMOE
      GO TO 40
 20  CONTINUE
      IF (IDRUN .NE. 2) GO TO 30
      RMOR=0
      NORMAL=0
      WRITE(*,110)
      READ(*,*) EMOE
      GO TO 50
 30  CONTINUE
      IF (IDRUN .NE. 1) GO TO 10
      EMOE=0
 40  CONTINUE
      IF (LOOP .EQ. 1) GO TO 45
      WRITE(*,120)
120  FORMAT(1X, 'DO YOU WANT TO NORMALIZE MOR (YES=1, NO=O)?')
      READ(*,*) NORMAL
      IF(NORMAL .EQ. 0) Go TO 45
      IF(NORMAL .NE. 1) GO TO 40
      WRITE(*,125)
125  FORMAT(1X, 'WHAT IS THE MEAN MOR OF THE 2x4 SELECT STRUCTURAL PIECE
    XS AT 15%?')
      READ(*,*) SPMEAN
 45  CONTINUE
      WRITE(*,115)
115  FORMAT(1X, 'PLEASE ENTER MOR VALUE')
      READ(*,*) RMOR
 50  CONTINUE
      WRITE(*,130)
130  FORMAT(1X, 'WHAT IS ITS PERCENT MOISTURE CONTENT (EXAMPLE 10.0)?')
      READ(*,*) AMC
      WRITE(*,135)
135  FORMAT(1X, 'WHAT PERCENT MOISTURE CONTENT DO YOU WANT IT ADJUSTED T
```

```
    X0?')
     READ(*,*)  TMC
     CALL MCADJ (RMOR, EMOE, AMC, TMC, NORMAL, SPMEAN, RADJ, EADJ, RNADJ)
     WRITE(*,140)
140  FORMAT( '1
    X           ADJUSTED ' )
     WRITE(*,145)
145 FORMAT(1X, '                                        INITIAL          ADJUSTED
    X            NORMALIZED')
     WRITE(*,150)
150 FORMAT(1X, '                                         VALUE            VALUE
    X              VALUE')
     WRITE(*,155)
155 FORMAT( '
    X       ____________')
     IF  (IDRUN  .EQ.  2)  GO  TO  60
     WRITE(*,160)
160  FORMAT(1X,  '                MOR')
     WRITE(*, 165) RMOR, RADJ, RNADJ
165  FORMAT(1X,  '           (1,000  PSI)            ',F6.3,'           ',F6.3
    X,'              ',F6.3)
     WRITE(*,170)
170  FORMAT(1X,  '  ')
 60  CONTINUE
     IF  (IDRUN  .EQ.  1)  GO  TO  70
     WRITE(*,175)
175 FORMAT(1X, '                MOE')
     WRITE(*,180)  EMOE,  EADJ
180 FORMAT(1X, '          (1,000,000  PSI)          ',F5.3,'            ',F5
    X.3)
     WRITE(*,170)
 70 CONTINUE
     WRITE(*,185)  AMC,  TMC
185 FORMAT(1X, '                 MC                ',F5.1,'            ',F5.1
    X)
     IF  (IDRUN  .EQ.  2)  GO  TO  80
     IF  ((RADJ  .LE.  0.1)  .OR.  ((NORMAL  .EQ.  1)  .AND.
    X     (RNADJ  .LE.  0.1)))  THEN
        WRITE(*,170)
        WRITE(*,170)
        WRITE(*,188)
188     FORMAT (1X, ' IF THE ADJUSTED MOR OR ADJUSTED NORMALIZED MOR IS L
    XESS THAN OR EQUAL ' )
        WRITE(*,189)
189     FORMAT (1X, ' TO 0.1 THEN THERE IS PROBABLY AN ERROR IN THE INPUT
    X DATA ' )
       ELSE
     END IF
 80  CONTINUE
     WRITE(*,l70)
     WRITE(*,170)
     WRITE(*,170)
     WRITE(*,190)
190 FORMAT (1X, 'DO YOU WISH To CONTINUE THE SAME PROCESSING FOR OTHER
```

```
     X  VALUES  (YES=1,  NO=0)?')
      READ(*,*)  LOOP
      IF  (LOOP  .EQ.  1)  GO  TO  15
      STOP
      END
      SUBROUTINE MCADJ(RMOR, EMOE, AMC, TMC, NORMAL, SPMEAN, RADJ, EADJ, RNADJ)
C
C         SUBROUTINE  TO  ADJUST  STRENGTH  PROPERTIES  FOR  MOISTURE  CONTENT
C
C         INPUT:'
C
C            RMOR     - MOR OF SPECIMEN
C            EMOE     - MOE OF SPECIMEN
C            AMC       - ACTUAL  MOISTURE  CONTENT  OF  SPECIMEN
C            TMC      - TARGET  MOISTURE  CONTENT
C            NORMAL -  1 IF NORMALIZATION IS DESIRED
C                      0 IF NORMALIZATION IS NOT DESIRED
C            SPMEAN -  MEAN  OF  2x4  SELECT  STRUCTURAL  PIECES  AT  15%  MC  IF
C                           NORMALIZATION  IS  DESIRED
C                      UNDEFINED  IF  NORMALIZATION  IS  NOT  DESIRED
C
C         OUTPUT:
C
C            RADJ     -  ADJUSTED  MOR  VALUE  (UNNORMALIZED)
C            EADJ     -  ADJUSTED  MOE  VALUE
C            RNADJ    -  ADJUSTED  MOR  VALUE  (NORMALIZED)  -  IF  DESIRED
C
C         CHECK  TO  SEE  IF  VARIABLES  ARE  READ  IN  TO  CORRECT  SCALE
C
      IF(RMOR  .GE.  0.0)  GO  TO  10
      WRITE(*,100)  RMOR
  100 FORMAT(1X, 'YOUR MOR VALUE ',F7.3,' IS NEGATIVE AND NOT ALLOWED BY
     XTHIS  PROGRAM'  )
      STOP
   10 CONTINUE
      IF(NORMAL  .EQ.  0)  GO  TO  15
      IF(SPMEAN  .GE.  0.0)  GO  TO  15
      WRITE(*,105)  SPMEAN
  105 FORMAT(1X, 'YOUR MEAN MOR VALUE ',F7.3,' OF THE 2X4 SELECT STRUCTU
     XRAL PIECES FOR'  )
      WRITE(*,110)
110   FORMAT(1X, 'NORMALIZATION IS NEGATIVE AND NOT ALLOWED BY THIS PROG
     XRAM' )
      STOP
   15 CONTINUE
      IF(EMOE  .GE.  0.0)  GO  TO  20
      WRITE(*,115)  EMOE
  115 FORMAT(1X, 'YOUR MOE VALUE ',F6.3,' IS NEGATIVE AND NOT ALLOWED BY
     X THIS PROGRAM' )
      STOP
   20 CONTINUE
      IF(RMOR  .GE.  0.0  .AND.  RMOR  .LE.  50.0)  GO  TO  25
      RMOR=RMOR/1000.0
   25 CONTINUE
```

```
      IF (NORMAL .EQ. 0) GO TO 30
      IF(SPMEAN .GE. 0.0 .AND. SPMEAN .LE. 50.0) GO TO 30
      SPMEAN=SPMEAN/1000.0
   30 CONTINUE
      IF(EMOE .GE. 0.0 .AND. EMOE .LT. 10.0) GO TO 35
      EMOE=EMOE/1000000.0
   35 CONTINUE
C
C          SETTING INITIAL VALUES
C
      XMEAN=10.12045
      IF(AMC .LT. 8.0) AMC=8.0
      IF(AMC .GT. 23.0) AMC=23.0
      IF(TMC .GE. 8.0 .AND. TMC .LE. 23.0) Go To 40
      WRITE(*,120) TMC
  120 FORMAT(1X, 'YOUR TARGET MOISTURE CONTENT OF ',F6.1,' IS OUTSIDE THE
     X LIMITS OF 8 TO 23 ALLOWED BY THIS PROGRAM' )
      STOP
   40 CONTINUE
C
C          ADJUSTING THE DATA WITH THE CORRECT ADJUSTMENT TO TARGET MOISTURE
C
      CALL ECALC(EMOE, AMC, TMC, EADJ)
      IF(EMOE .LE. 0.0) EADJ=0.0
      CALL RCALC(RMOR, AMC, TMC, RADJ)
      IF (RADJ .LT. 0.1) RADJ=0.1
      IF (RMOR .LE. 0.0) RADJ=0.0
      IF(NORMAL .NE. 1) GO TO 45
C
C          NORMALIZING THE MODELS
C
      DIFF=RMOR-1.0
      ZMOR=(DIFF*(XMEAN/SPMEAN))+1.0
      IF(RMOR .LE. 1.0) ZMOR=RMOR
C
C          NORMALIZING TO TARGET MOISTURE CONTENT
C
      CALL RCALC(ZMOR, AMC, TMC, ADJRN)
C
C          TAKING OUT THE NORMALIZATION FOR TARGET MOISTURE CONTENT
C
      DIFF2=ADJRN-1.0
      RNADJ=(DIFF2*(SPMEAN/XMEAN))+1.0
      IF(ZMOR .LE. 1.0) RNADJ=ADJRN
      IF(RMOR .LE. 0.0) RNADJ=0.0
   45 CONTINUE
      RETURN
      END
      SUBROUTINE RCALC(RMOR, AMC, TMC, RADJ)
C
C          SUBROUTINE To FIT A FIXED QUADRATIC SURFACE (CUBIC)
C               ASSUMPTIONS : DF MC 23
C                             SP MC 23
C                             PERCENTILES DON'T MATCH
```

```
C
      COMMON C
      DIMENSION B(8) ,C(4)
      DATA B /-.45336002443, .37073911,-.047331957,
     X .0013200499,
     X .01348986358,-.01083274,.0012329926,-0.000033199128/
C
C         CHECK IF RMOR IS POSITIVE
C
      IF (RMOR .GT. 0.0) GO TO 10
      RADJ=0.0
      GO TO 100
   10 CONTINUE
C
C         SETTING UP THE CUBIC EQUATION C1 + C2*R + C3*(R**2) + C4*(R**3)
C         FOR WHICH ROOTS ARE DESIRED
C
      C(1)=RMOR+(B(1)*(15.0-AMC))+(B(5)*((15.0*15.0)-(AMC*AMC)))
      C(2)=(B(2)*(15.0-AMC))+(B(6)*((15.0*15.0)-(AMC*AMC)))-1.0
      C(3)=(B(3)*(15.0-AMC))+(B(7)*((15.0*15.0)-(AMC*AMC)))
      C(4)=(B(4)*(15.0-AMC))+(B(8)*((15.0*15.0)-(AMC*AMC)))
C
C         FINDING THE ROOTS
C
      IF (AMC .EQ. 15.0) THEN
         R15=RMOR
        ELSE
         CALL CUBIC (C,R15,RMOR)
      END IF
C
C         CONSTRUCTING AN UPPER BOUND
C
      IF (R15 .GT. 13.0) R15=13.0
C
C         CONSTRUCTING A LOWER BOUND
C
      IF (RMOR .LT. 1.0) R15=1.488
      IF (R15 .LT. 1.488) R15=1.488
C
C         FINDING MOR AT TMC
C
      B115=B(1)+(B(2)*R15)+(B(3)*R15*R15)+(B(4)*R15*R15*R15)
      B1115=B(5)+(B(6)*R15)+(B(7)*R15*R15)+(B(8)*R15*R15*R15)
      Y=RMOR+(B115*(TMC-AMC))+(B1115*(TMC*TMC-AMC*AMC))
C
      RADJ=Y
      IF(RADJ .LT. 0.1) RADJ=0.1
  100 CONTINUE
      RETURN
      END
      SUBROUTINE CUBIC (C,R15,RMOR)
C
C     SUBROUTINE TO FIND ROOTS IN THE CUBIC EQUATION:
C            (C(4)*R15*R15*R15)+(C(3)*R15*R15)+(C(2)*R15)+C(1)=0
```

```
C       USING NOTATION AND METHODS DISCUSSED IN THE C.R.C. STANDARD
C       MATHEMATICAL  TABLES
C
        DIMENSION  X(3),  C(4),  Y(3)
C
C       THE EQUATION MAY BE WRITTEN:
C       (R15*R15*R15)+((C(3)/C(4)*(R15*R15))+((C(2)/C(4))*R15)+(C(1)/C(4))=0
C       OR WHEN CHANGED TO C.R.C. NOTATION AS:
C               ((Y*Y*Y)+(P*(Y*Y))+(Q*Y)+R=0  WHEN
C       Y IS USED FOR R15 AND
C
        P=C(3)/C(4)
        Q=C(2)/C(4)
        R=C(1)/C(4)
C
C       IF WE SUBSTITUTE Y=X-(P/3) FOR Y, THE EQUATION BECOMES
C               (X*X*X)+(A*X)+B=0
C
C       FOR SOLUTION OF THE ROOTS:
C
        A=((3*Q)-(P*P))/3.
        B=(2*(P*P*P)-(9*P*Q)+(27*R))/27
        Z=((B*B)/4)+((A*A*A)/27)
C
C       WHETHER THE VALUE OF THE TERM Z IS POSITIVE, ZERO, OR NEGATIVE
C       DETERMINES HOW MANY REAL ROOTS THERE ARE AND HOW THE ROOTS
C       ARE CALCULATED.   THE CALCULATED ROOTS WILL BE PLACED IN THE ARRAY Y
C       NOTE:   THE "CUBE ROOT OF ANY VALUE N"  =  EXP((LOG(N))/3)  WITH  SIGN
C       RETURNED
C
C       IF Z IS POSITIVE THERE IS ONE ROOT WHICH IS CALCULATED AND PLACED IN
C       Y(1)
C
        IF  (Z  .GT.  0.0)  THEN
          NROOT=1
          ZA=(-(B/2)+SQRT(Z))
          ZB=(-(B/2)-SQRT(Z))
          IF  (ZA  .EQ.  0)  CA=0
          IF  (ZB  .EQ.  0)  CB=0
          IF  (ZA  .GT.  0)  CA=EXP((LOG(ZA))/3)
          IF  (ZB  .GT.  0)  CB=EXP((LOG(ZB))/3)
          IF  (ZA  .LT.  0)  CA=-EXP((LOG(-1*ZA))/3)
          IF  (ZB  .LT.  0)  CB=-EXP((LOG-(-1*ZB))/3)
          X(1)=CA+CB
C
C       THIS IS THE SOLUTION IN TERMS OF X
C       FOR A SOLUTION IN TERMS OF Y, WE MUST SUBSTITUTE Y=X-(P/3)
C
          Y(1)=X(1)-(P/3)
          R15=Y(1)
         ELSE=
          NROOT=3
C
C       IF Z IS 0 THERE ARE THREE ROOTS, TWO OF WHICH ARE EQUAL
```

```
C
            IF (Z .EQ. 0.0) THEN
              ZN=-(B/2)
              IF (ZN .EQ. 0) CA=0
              IF (ZN .GT. 0) CA=EXP((LOG(ZN))/3)
              IF (ZN .LT. 0) CA=-EXP((LOG(-1*ZN))/3)
              CB=CA
              X(1)=CA+CB
              X(2)=-(CA+CB)/2
              X(3)=X(2)
             ELSE
C
C     IF Z IS NEGATIVE THERE ARE THREE ROOTS WHICH ARE CALCULATED USING
C     A TRIGONOMETRIC SOLUTION
C
              THETA=ACOS((-B/2)/SQRT((-A*A*A)/27))
              X(1)=2*(SQRT(-A/3))*COS(THETA/3)
              X(2)=2*(SQRT(-A/3))*COS((THETA/3)+2.0944)
              X(3)=2*(SQRT(-A/3))*COS((THETA/3)+4.1888)
            END IF
      END IF
C
C     IN THE CASE OF MULTIPLE ROOTS, WE NOW HAVE THE SOLUTIONS IN TERMS OF X
C     FOR A SOLUTION IN TERMS OF Y, WE MUST SUBSTITUTE Y=X-(P/3)
C
      IF (NROOT :EQ. 3) THEN
        Y(1)=X(1)-(P/3)
        Y(2)=X(2)-(P/3)
        Y(3)=X(3)-(P/3)
C
C     IN THE CASE OF MULTIPLE ROOTS, THE ROOT CLOSEST TO RMOR WILL BE USED
C
        IF (ABS(RMOR-Y(1)) .GT. ABS(RMOR-Y(2))) Y(1)=Y(2)
        IF (ABS(RMOR-Y(1)) .GT. ABS(RMOR-Y(3))) Y(1)=Y(3)
        R15=Y(1)
       ELSE
      END IF
      RETURN
      END
C
      SUBROUTINE ECALC(EMOE, AMC, TMC, EADJ)
C
C     SUBROUTINE To CALCULATE MOE OF SPECIMEN AT TARGET MOISTURE
C
C         CONSTANT PERCENTAGE ADJUSTMENT
C             ASSUMPTIONS: DF MC 23
C                          SP MC 23
C
      A=1.8566
      B=-.023722
      PCT=((A+B*TMC)/(A+B*AMC))
      EADJ=EMOE*PCT
      RETURN
      END
```

2. Subroutine TCALC for UTS

```
      SUBROUTINE TCALC(UTS, AMC, TMC, TADJ)
C
C         SUBROUTINE TO FIT A FIXED QUADRATIC SURFACE (CUBIC)
C             ASSUMPTIONS : DF MC 23
C                           SP MC 23
C                           PERCENTILES DON'T MATCH
C
      COMMON C
      DIMENSION  B(8),C(4)
      DATA  B  /-.18947228958,0.29393506,-0.054178160,
     X 0.0031627702,
     X 0.00585499434,-0.00843352,0.0014837455,-0.000088102328/
C
C         CHECK IF UTS IS POSITIVE
C
      IF (UTS .GT. 0.0) GO TO 10
      TADJ=0.0
      GO TO 100
   10 CONTINUE
C
C         SETTING UP THE CUBIC EQUATION C1 + C2*R + C3*(R**2) + C4*(R**3)
C         FOR WHICH ROOTS ARE DESIRED
C
      C(1)=UTS+(B(1)*(15.0-AMC))+(B(5)*((15.0*15.0)-(AMC*AMC)))
      C(2)=(B(2)*(15.0-AMC))+(B(6)*((15.0*15.0)-(AMC*AMC)))-1.0
      C(3)=(B(3)*(15.0-AMC))+(B(7)*((15.0*15.0)-(AMC*AMC)))
      C(4)=(B(4)*(15.0-AMC))+(B(8)*((15.0*15.0)-(AMC*AMC)))
C
C         FINDING THE ROOTS
C
      IF (AMC .EQ. 15.0) THEN
         T15=UTS
        ELSE
         CALL CUBIC (C, T15, UTS)
      END IF
C
C         CONSTRUCTING AN UPPER BOUND
C
      IF (T15 .GE. 10.0) T15=10.0
      IF (UTS .GE. 10.0) T15=10.0
C
C         CONSTRUCTING A LOWER BOUND
C
      IF (T15 .LE. 0.9) T15=0.9
      IF (UTS .LE. 0.3) T15=0.9
C
```

```
C       FINDING UTS AT TMC
C
        B115=B(1)+(B(2)*T15)+(B(3)*T15*T15)+(B(4)*T15*T15*T15)
        B1115=B(5)+(B(6)*T15)+(B(7)*T15*T15)+(B(8)*T15*T15*T15)
        Y=UTS+(B115*(TMC-AMC))+(B1115*(TMC*TMC-AMC*AMC))
C
C          FANNING THE BOTTOM
C
        IF (T15 .GT. 0.9) GO TO 50
        CHANGE=Y-UTS
        TMCB=15.0
        UTS2=0.9
        TATAMC=UTS2+(B115*(AMC-TMCB))+(B1115*(AMC*AMC-TMCB*TMCB))
        CHANGE=CHANGE*(UTS/TATAMC)
        Y=UTS+CHANGE
   50   CONTINUE
C
C
        TADJ=Y
  100   CONTINUE
        RETURN
        END
```

3. Subroutine STCALC for UCS

```
        SUBROUTINE STCALC(STRESS,AMC,TMC,STADJ)
C
C          SUBROUTINE TO FIT A FIXED QUADRATIC SURFACE (CUBIC)
C             ASSUMPTIONS :  DF MC 23
C                            SP MC 23
C                            PERCENTILES DON'T MATCH
C
        COMMON C
        DIMENSION B(8),C(4)
        DATA B /0.173389,0.137645,-0.0875026,0.00733659,
       X -0.0036906,-0.004534,0.00224092,-0.000189583/
C
C          CHECK IF STRESS IS POSITIVE
C
        IF (STRESS .GT. 0.0) GO TO 10
        STADJ=0.0
        GO TO 100
   10   CONTINUE
C
C          SETTING UP THE CUBIC EQUATION C1 + C2*R + C3*(R**2) + C4*(R**3)
C          FOR WHICH ROOTS ARE DESIRED
C
        C(1)=STRESS+(B(1)*(15.0-AMC))+(B(5)*((15.0*15.0)-(AMC*AMC)))
        C(2)=(B(2)*(15.0-AMC))+(B(6)*((15.0*15.0)-(AMC*AMC)))-1.0
        C(3)=(B(3)*(15.0-AMC))+(B(7)*((15.0*15.0)-(AMC*AMC)))
```

```
      C(4)=(B(4)*(15.0-AMC))+(B(8)*((15.0*15.0)-(AMC*AMC)))
C
C        FINDING THE ROOTS
C
      IF (AMC .EQ. 15.0) THEN
         ST15=STRESS
        ELSE
         CALL CUBIC (C,ST15,STRESS)
      END IF
C
C        CONSTRUCTING AN UPPER BOUND
C
      IF (ST15 .GT. 6.393) ST15=6.393
      IF (STRESS .GE. 10.0) ST15=6.393
C
C        CONSTRUCTING A LOWER BOUND
C
      IF (ST15 .LE. 2.5) ST15=2.5
C
C     FINDING STRESS AT TMC
C
      B115=B(1)+(B(2)*ST15)+(B(3)*ST15*ST15)+(B(4)*ST15*ST15*ST15)
      B1115=B(5)+(B(6)*ST15)+(B(7)*ST15*ST15)+(B(8)*ST15*ST15*ST15)
      Y=STRESS+(B115*(TMC-AMC))+(B1115*(TMC*TMC-AMC*AMC))
C
C        FANNING THE BOTTOM
C
      IF (ST15 .GT. 2.5) GO TO 50
      CHANGE=Y-STRESS
      TMCB=15.0
      STRESS2=2.5
      SATAMC=STRESS2+(B115*(AMC-TMCB))+(B1115*(AMC*AMC-TMCB*TMCB))
      CHANGE=CHANGE*(STRESS/SATAMC)
      Y=STRESS+CHANGE
   50 CONTINUE
C
C
      STADJ=Y
  100 CONTINUE
      RETURN
      END
```


☆U. S. GOVERNMENT PRINTING OFFICE: 1990/743-044/00015